

irf describe — Describe an IRF file

[Description](#)
[Options](#)[Quick start](#)
[Remarks and examples](#)[Menu](#)
[Stored results](#)[Syntax](#)
[Also see](#)

Description

`irf describe` describes the specification of the estimation command and the specification of the IRF used to create the IRF results that are saved in an IRF file.

Quick start

Short summary of all IRF results in the active IRF file

```
irf describe
```

Summary of model and IRF specification for `irf1` in the active IRF file

```
irf describe irf1
```

As above, but for `irf1` in IRF file `myirf.irf`

```
irf describe irf1, using(myirf)
```

As above, and also set `myirf.irf` as the active IRF file

```
irf describe irf1, set(myirf)
```

Note: `irf` commands can be used after `var`, `svar`, `vec`, `arma`, or `arfima`; see [\[TS\] var](#), [\[TS\] var svar](#), [\[TS\] vec](#), [\[TS\] arma](#), or [\[TS\] arfima](#).

Menu

Statistics > Multivariate time series > Manage IRF results and files > Describe IRF file

Syntax

```
irf describe [irf_resultslist] [, options]
```

<i>options</i>	Description
<code>set(<i>filename</i>)</code>	make <i>filename</i> active
<code>using(<i>irf_filename</i>)</code>	describe <i>irf_filename</i> without making active
<code>detail</code>	show additional details of IRF results
<code>variables</code>	show underlying structure of the IRF dataset

Options

`set(filename)` specifies the IRF file to be described and set; see [TS] `irf set`. If *filename* is specified without an extension, `.irf` is assumed.

`using(irf_filename)` specifies the IRF file to be described. The active IRF file, if any, remains unchanged. If *irf_filename* is specified without an extension, `.irf` is assumed.

`detail` specifies that `irf describe` display detailed information about each set of IRF results. `detail` is implied when *irf_resultslist* is specified.

`variables` is a programmer's option; additionally displays the output produced by the `describe` command.

Remarks and examples

[stata.com](http://www.stata.com)

If you have not read [TS] `irf`, please do so.

`irf describe` specified without *irf_resultslist* provides a short summary of the model used to create each set of results in an IRF file. If *irf_resultslist* is specified, then `irf describe` provides details of the model specification and the IRF specification used to create each set of IRF results. If `set()` or `using()` is not specified, the IRF results of the active IRF file are described.

► Example 1

```
. use http://www.stata-press.com/data/r15/lutkepohl2
(Quarterly SA West German macro data, Bil DM, from Lutkepohl 1993 Table E.1)
. var dln_inv dln_inc dln_consump if qtr<=tq(1978q4), lags(1/2) dfk
(output omitted)
```

We create three sets of IRF results:

```
. irf create order1, set(myirfs, replace)
(file myirfs.irf created)
(file myirfs.irf now active)
(file myirfs.irf updated)
. irf create order2, order(dln_inc dln_inv dln_consump)
(file myirfs.irf updated)
. irf create order3, order(dln_inc dln_consump dln_inv)
(file myirfs.irf updated)
. irf describe
Contains irf results from myirfs.irf (dated 30 Mar 2017 23:01)
```

irfname	model	endogenous variables and order (*)
order1	var	dln_inv dln_inc dln_consump
order2	var	dln_inc dln_inv dln_consump
order3	var	dln_inc dln_consump dln_inv

(*) order is relevant only when model is var

The output reveals the order in which we specified the variables.

```
. irf describe order1
irf results for order1
Estimation specification
  model: var
  endog: dln_inv dln_inc dln_consump
  sample: quarterly data from 1960q4 to 1978q4
  lags: 1 2
  constant: constant
  exog: none
  exogvars: none
  exlags: none
  varcns: unconstrained

IRF specification
  step: 8
  order: dln_inv dln_inc dln_consump
  std error: asymptotic
  reps: none
```

Here we see a summary of the model we fit as well as the specification of the IRFs.

Stored results

`irf describe` stores the following in `r()`:

Scalars

<code>r(N)</code>	number of observations in the IRF file
<code>r(k)</code>	number of variables in the IRF file
<code>r(width)</code>	width of dataset in the IRF file
<code>r(N_max)</code>	maximum number of observations
<code>r(k_max)</code>	maximum number of variables
<code>r(widthmax)</code>	maximum width of the dataset
<code>r(changed)</code>	flag indicating that data have changed since last saved

Macros

<code>r(_version)</code>	version of IRF results file
<code>r(irfnames)</code>	names of IRF results in the IRF file
<code>r(irfname_model)</code>	<code>var</code> , <code>sv var</code> , <code>lr var</code> , or <code>vec</code>
<code>r(irfname_order)</code>	Cholesky order assumed in IRF estimates
<code>r(irfname_exog)</code>	exogenous variables, and their lags, in VAR or underlying VAR
<code>r(irfname_exogvar)</code>	exogenous variables in VAR or underlying VAR
<code>r(irfname_constant)</code>	constant or noconstant
<code>r(irfname_lags)</code>	lags in model
<code>r(irfname_exlags)</code>	lags of exogenous variables in model
<code>r(irfname_tmin)</code>	minimum value of timevar in the estimation sample
<code>r(irfname_tmax)</code>	maximum value of timevar in the estimation sample
<code>r(irfname_timevar)</code>	name of <code>tsset</code> timevar
<code>r(irfname_tsfmt)</code>	format of timevar in the estimation sample
<code>r(irfname_varcns)</code>	unconstrained or colon-separated list of constraints placed on VAR coefficients
<code>r(irfname_svarcns)</code>	"," or colon-separated list of constraints placed on SVAR coefficients
<code>r(irfname_step)</code>	maximum step in IRF estimates
<code>r(irfname_stderror)</code>	asymptotic, <code>bs</code> , <code>bsp</code> , or <code>none</code> , depending on type of standard errors specified to <code>irf create</code>
<code>r(irfname_reps)</code>	"," or number of bootstrap replications performed
<code>r(irfname_version)</code>	version of IRF file that originally held <code>irfname</code> IRF results
<code>r(irfname_rank)</code>	"," or number of cointegrating equations
<code>r(irfname_trend)</code>	"," or <code>trend()</code> specified in <code>vec</code>
<code>r(irfname_veccns)</code>	"," or constraints placed on VECM parameters
<code>r(irfname_sind)</code>	"," or normalized seasonal indicators included in <code>vec</code>

Also see

[TS] [irf](#) — Create and analyze IRFs, dynamic-multiplier functions, and FEVDs

[TS] [var intro](#) — Introduction to vector autoregressive models

[TS] [vec intro](#) — Introduction to vector error-correction models